Question 4: MUD-bGone

a) Formulate the MUD-bGone problem using a network flow to maximize profits. Solve using AMPL and clearly explain the optimal solution.

The following



Let’s step through its components.

Nodes 1-4 represent the product supply movement at periods 1 through 4. Node 5 represents the final inventory after period 4. Nodes 6-7 represent the raw materials that can be purchased at each period to increase product supply in the following period. Note that I assume the ratio of gallons of raw material to gallons of extra product remains constant (500 gallons of base material to 225 gallons of product) which means .45 gallons of product are generated per gallon of raw material. The network is supplied with 2500 gallons of product at period 1 and demands 2500 gallons of product reach period 5. The arcs between period nodes have multipliers of .94 to represent the 6% loss of product in inventory between periods.

Node 10 represents total cash generated by the system. The arcs between nodes 1-4 and 10 represent cash generated by selling product, as the multipliers along these arcs are the prices in dollars per gallon at each respective period. The arcs from node 10 to 6-9 represent the cash required to purchase raw material. The multipliers along these arcs must have units of gallons per dollar, so they are given by the inverse of the cost of raw material per gallon. Node 11 is a dummy node where excess cash can be dumped, hence the multiplier of 0 on the arc between nodes 10 and 11. The cost along this arc is 1, representing a $1 profit for every unit of flow along it.

The above network flow was then transferred into AMPL and solved with CPLEX.

Model file: This is modified from the gmcnfp model on Canvas.

```
# AMPL model for the Minimum Cost Network Flow Problem
#
# By default, this model assumes that b[i] = 0, c[i,j] = 0,
# l[i,j] = 0 and u[i,j] = Infinity.
#
# Parameters not specified in the data file will get their default values.

options solver cplex;

set NODES;                        # nodes in the network
set ARCS within {NODES, NODES};   # arcs in the network

param b {NODES} default 0;        # supply/demand for node i
param c {ARCS}  default 0;        # cost of one of flow on arc(i,j)
param l {ARCS}  default 0;        # lower bound on flow on arc(i,j)
param u {ARCS}  default Infinity; # upper bound on flow on arc(i,j)
param mu {ARCS} default 1;       # multiplier on arc(i,j) -- if one unit leaves i, mu[i,j] units arrive

var x {ARCS};                     # flow on arc (i,j)

maximize cost: sum{(i,j) in ARCS} c[i,j] * x[i,j];  #objective: maximize arc flow cost

# Flow Out(i) - Flow In(i) = b(i)

subject to flow_balance {i in NODES}:
sum{j in NODES: (i,j) in ARCS} x[i,j] - sum{j in NODES: (j,i) in ARCS} mu[j,i] * x[j,i] = b[i];

subject to capacity {(i,j) in ARCS}: l[i,j] <= x[i,j] <= u[i,j];

#load in data and solve
data HW3/HW3_Q4a_data.txt;
solve;

#display optimal flow along each arc
printf "The following flows (from arc i to j) are optimal for our problem:\n";
display x;

printf "The total profit (in dollars) resulting from this network flow is:\n";
display cost;
```

Data File: Too long to include nicely in this pdf. It is attached in the .zip file.

Output:

```
ampl: model HW3/model_Q4a.txt
CPLEX 20.1.0.0: sensitivity
CPLEX 20.1.0.0: optimal solution; objective 474360.5556
2 dual simplex iterations (0 in phase I)

suffix up OUT;
suffix down OUT;
suffix current OUT;
The following flows (from arc i to j) are optimal for our problem:
x :=
1  2         0
1  10     2500
2  3       375
2  10     3000
3  4         0
3  10     4402.5
4  5         0
4  10     3825
6  2      7500
7  3      9000
8  4      8500
9  5      5555.56
10 6     18750
10 7     22500
10 8     25500
10 9     19444.4
10 11   474361
;

The total profit (in dollars) resulting from this network flow is:
cost = 474361
```

This output suggests that the maximum profit possible is $474,360.56.

b) What is the per-gallon value of having additional supply of raw chemical base available in period 2? This marginal value is valid in what range (i.e., for what upper and lower limits of period 2 supply)?

The model file from part A was slightly modified to include CPLEX's sensitivity analysis. Since we want to know impact of having additional supply of raw chemical base in period 2, we are really looking at the impact of the upper bound on flow between nodes 7 and 3. So we split the flow constraint in two (upper and lower) and examine the shadow price of the upper bound between nodes 7 and 3.

Model file:

```
# AMPL model for the Minimum Cost Network Flow Problem
#
# By default, this model assumes that b[i] = 0, c[i,j] = 0,
# l[i,j] = 0 and u[i,j] = Infinity.
#
# Parameters not specified in the data file will get their default values.

reset;
options solver cplex;
options cplex_options 'sensitivity';

set NODES;                        # nodes in the network
set ARCS within {NODES, NODES};   # arcs in the network

param b {NODES} default 0;        # supply/demand for node i
param c {ARCS}  default 0;        # cost of one of flow on arc(i,j)
param l {ARCS}  default 0;        # lower bound on flow on arc(i,j)
param u {ARCS}  default Infinity; # upper bound on flow on arc(i,j)
param mu {ARCS} default 1;       # multiplier on arc(i,j) -- if one unit leaves i, mu[i,j] units arrive

var x {ARCS};                     # flow on arc (i,j)

maximize cost: sum{(i,j) in ARCS} c[i,j] * x[i,j];  #objective: maximize arc flow cost

# Flow Out(i) - Flow In(i) = b(i)

subject to flow_balance {i in NODES}:
sum{j in NODES: (i,j) in ARCS} x[i,j] - sum{j in NODES: (j,i) in ARCS} mu[j,i] * x[j,i] = b[i];

subject to l_capacity {(i,j) in ARCS}: l[i,j] <= x[i,j];
subject to u_capacity {(i,j) in ARCS}: x[i,j] <= u[i,j];

#load in data and solve
data HW3/HW3_Q4a_data.txt;
solve;

#display answers to question b
printf "The per-gallon value of additional raw chemical base in period 2 is given by:\n";
display u_capacity[7,3];
printf "This marginal value is valid within the range: (these values are bugged) \n";
display u_capacity[7,3].down, u_capacity[7,3].up;
```

Data File: Same as in part A.

Output:

```
ampl: model HW3/model_Q4b.txt
CPLEX 20.1.0.0: sensitivity
CPLEX 20.1.0.0: optimal solution; objective 474360.5556
2 dual simplex iterations (0 in phase I)

suffix up OUT;
suffix down OUT;
suffix current OUT;
The per-gallon value of additional raw chemical base in period 2 is given by:
u_capacity[7,3] = 16.4

This marginal value is valid within the range:
u_capacity[7,3].down = 0
u_capacity[7,3].up = 0

ampl: model HW3/model_Q4b.txt
CPLEX 20.1.0.0: sensitivity
CPLEX 20.1.0.0: optimal solution; objective 474360.5556
2 dual simplex iterations (0 in phase I)

suffix up OUT;
suffix down OUT;
suffix current OUT;
The per-gallon value of additional raw chemical base in period 2 is given by:
u_capacity[7,3] = 16.4

This marginal value is valid within the range: (these values are bugged)
u_capacity[7,3].down = 0
u_capacity[7,3].up = 0
```

Note the shadow price of 16.4. That is, per gallon of additional raw material available in period two, we expect to see an increase in profit of $16.40. The marginal values were bugged, but a brute force search yielded a range of feasibility between 3413 gallons at a minimum and 29218 gallons at a maximum.

c) How sensitive is the optimal solution to the price/gallon of MUD-bGone in period 2?

The product's price at period 2 is represented by the multiplier between node 2 and node 10. To test the sensitivity of the optimal solution (profit) to this price, I resolved the model from part A, two additional times with the multiplier incremented by one each time in the data file.

Output:

```
ampl: model HW3/model_Q4c.txt;
CPLEX 20.1.0.0: sensitivity
CPLEX 20.1.0.0: optimal solution; objective 474360.5556
2 dual simplex iterations (0 in phase I)

suffix up OUT;
suffix down OUT;
suffix current OUT;

The price (in dollars) for MUD-bGone in this run is:
mu[2,10] = 40

The total profit (in dollars) resulting from this network flow is:
cost = 474361

ampl: model HW3/model_Q4c.txt;
CPLEX 20.1.0.0: sensitivity
CPLEX 20.1.0.0: optimal solution; objective 477360.5556
2 dual simplex iterations (0 in phase I)

suffix up OUT;
suffix down OUT;
suffix current OUT;

The price (in dollars) for MUD-bGone in this run is:
mu[2,10] = 41

The total profit (in dollars) resulting from this network flow is:
cost = 477361

ampl: model HW3/model_Q4c.txt;
CPLEX 20.1.0.0: sensitivity
CPLEX 20.1.0.0: optimal solution; objective 480360.5556
2 dual simplex iterations (0 in phase I)

suffix up OUT;
suffix down OUT;
suffix current OUT;

The price (in dollars) for MUD-bGone in this run is:
mu[2,10] = 42

The total profit (in dollars) resulting from this network flow is:
cost = 480361

ampl:
ampl:
ampl: display 480361-477361;
480361 - 477361 = 3000

ampl: display 480361 - 474361;
480361 - 474361 = 6000
```

I then took the profits from each run, took their difference, and determined that a $1 increase in price corresponds to a $3000 increase in profit.